Einiges zur Laplace - Transformation, FS 2013, W.Quapp

```
Integrate[Exp[-s t] , {t, 0, Infinity}]
```

$\text{ConditionalExpression}\left[\frac{1}{s}, \text{Re}[s] > 0\right]$

```
LaplaceTransform[1, t, s]
```

$\frac{1}{s}$

liste1 = $\{e^{a t}, e^{-3 t}, e^{5 t}\}$

$\{e^{a t}, e^{-3 t}, e^{5 t}\}$

```
Map[LaplaceTransform[#, t, s] &, liste1] /. Log[e] → 1
```

$\left\{\frac{1}{-a+s}, \frac{1}{3+s}, \frac{1}{-5+s}\right\}$

```
Clear[f]
```

```
f[t_] := -1 /; 0 ≤ t < 4
f[t_] := 1 /; 4 ≤ t
```

```
Plot[f[t], {t, 0, 8}, PlotStyle → Thick]
```




```
LaplaceTransform[f[t], t, s]
```

LaplaceTransform[f[t], t, s]

```
(* d.h. die 'zweifach' -
 definierte Funktion f kann LaplaceTransform nicht bearbeiten, Ausweg: *)
```

```
Clear[f]
f[t_] := UnitStep[t - 4] - UnitStep[4 - t]
```

```
LaplaceTransform[f[t], t, s]
```

$\frac{e^{-4 s}}{s} - \frac{1 - e^{-4 s}}{s}$

```
liste2 = {t^3, Sin[a t], Cos[a t]}
```

$\{t^3, \mathrm{Sin}[a\,t], \mathrm{Cos}[a\,t]\}$

```
Map[LaplaceTransform[#, t, s] &, liste2]
```

$\left\{\frac{6}{s^4}, \frac{a}{a^2+s^2}, \frac{s}{a^2+s^2}\right\}$

```
(*  Shift - Eigenschaft *)

LaplaceTransform[Exp[-2 t] Cos[t], t, s]
```

$\frac{2+s}{1+(2+s)^2}$

```
Clear[f]

ablei = Table[D[f[t], {t, n}], {n, 1, 5}]
```

$\{f'[t], f''[t], f^{(3)}[t], f^{(4)}[t], f^{(5)}[t]\}$

```
Map[LaplaceTransform[#, t, s] &, ablei]
```

$\{-f[0] + s\,\mathrm{LaplaceTransform}[f[t], t, s],$
$-s\,f[0] + s^2\,\mathrm{LaplaceTransform}[f[t], t, s] - f'[0],$
$-s^2\,f[0] + s^3\,\mathrm{LaplaceTransform}[f[t], t, s] - s\,f'[0] - f''[0],$
$-s^3\,f[0] + s^4\,\mathrm{LaplaceTransform}[f[t], t, s] - s^2\,f'[0] - s\,f''[0] - f^{(3)}[0],$
$-s^4\,f[0] + s^5\,\mathrm{LaplaceTransform}[f[t], t, s] - s^3\,f'[0] - s^2\,f''[0] - s\,f^{(3)}[0] - f^{(4)}[0]\}$

```
LaplaceTransform[t^n, t, s]
```

$s^{-1-n}\,\mathrm{Gamma}[1+n]$

```
Clear[f]
f[t_] := (3 t - 1)^3

LaplaceTransform[f[t], t, s]
```

$\frac{162}{s^4} - \frac{54}{s^3} + \frac{9}{s^2} - \frac{1}{s}$

```
LaplaceTransform[f'[t], t, s]
```

$9\left(\frac{18}{s^3} - \frac{6}{s^2} + \frac{1}{s}\right)$

```
Clear[f, a, k, n]

Liste = {1, t^n, e^(a t) , Sin[k t], Cos[k t], Sinh[k t], Cosh[k t], t^n e^(a t) ,
   Exp[a t] Sin[k t], Exp[a t] Cos[k t], Exp[a t] Sinh[k t],
   Exp[a t] Cosh[k t], DiracDelta''[t], DiracDelta '[t], DiracDelta[t]}
```

$\{1, t^n, e^{a\,t}, \mathrm{Sin}[k\,t], \mathrm{Cos}[k\,t], \mathrm{Sinh}[k\,t],$
$\mathrm{Cosh}[k\,t], e^{a\,t}\,t^n, e^{a\,t}\,\mathrm{Sin}[k\,t], e^{a\,t}\,\mathrm{Cos}[k\,t], e^{a\,t}\,\mathrm{Sinh}[k\,t],$
$e^{a\,t}\,\mathrm{Cosh}[k\,t], \mathrm{DiracDelta}''[t], \mathrm{DiracDelta}'[t], \mathrm{DiracDelta}[t]\}$

```
Map[{#, LaplaceTransform[#, t, s]} &, Liste] /. Log[e] → 1 // TableForm
```

1	$\frac{1}{s}$
t^n	s^{-1-n} Gamma[1 + n]
$e^{a\,t}$	$\frac{1}{-a+s}$
Sin[k t]	$\frac{k}{k^2+s^2}$
Cos[k t]	$\frac{s}{k^2+s^2}$
Sinh[k t]	$\frac{k}{-k^2+s^2}$
Cosh[k t]	$\frac{s}{-k^2+s^2}$
$e^{a\,t}\,t^n$	$(-a+s)^{-1-n}$ Gamma[1 + n]
$e^{a\,t}$ Sin[k t]	$\frac{k}{k^2+(a-s)^2}$
$e^{a\,t}$ Cos[k t]	$\frac{-a+s}{k^2+(a-s)^2}$
$e^{a\,t}$ Sinh[k t]	$\frac{k}{-k^2+(a-s)^2}$
$e^{a\,t}$ Cosh[k t]	$\frac{-a+s}{-k^2+(a-s)^2}$
DiracDelta''[t]	s^2
DiracDelta'[t]	s
DiracDelta[t]	1

```
InverseLaplaceTransform[2 / (s^2 + 4), s, t]
```

Sin[2 t]

```
s1 = InverseLaplaceTransform[2 / (s^2 + 2 s + 5), s, t]
```

$$-\frac{1}{2}\, i\, e^{(-1-2 i)\,t}\left(-1+e^{4 i t}\right)$$

```
ExpToTrig[s1 ] // FullSimplify
```

e^{-t} Sin[2 t]

```
(*  DGl mit LaplaceTransform *)
```

```
step1 = LaplaceTransform[y'[t] - 4 y[t] == Exp[4 t], t, s]
```

$$-4\,\text{LaplaceTransform}[y[t], t, s] + s\,\text{LaplaceTransform}[y[t], t, s] - y[0] = \frac{1}{-4+s}$$

```
step2 = step1 /. y[0] → 0
```

$$-4\,\text{LaplaceTransform}[y[t], t, s] + s\,\text{LaplaceTransform}[y[t], t, s] = \frac{1}{-4+s}$$

```
step3 = Solve[step2, LaplaceTransform[y[t], t, s]]
```

$$\left\{\left\{\text{LaplaceTransform}[y[t], t, s] \to \frac{1}{(-4+s)^2}\right\}\right\}$$

```
solution = InverseLaplaceTransform[step3[[1, 1, 2]], s, t]
```

$e^{4\,t}\,t$

```
(*  waere hier auch leicht mit DSolve ... gegangen !! *)

Plot[solution, {t, 0, 0.3}]
```




```
(* Der Plot zeigt, dass die AW erfuellt sind *)

(* next *)

step1 = LaplaceTransform[y''[t] + 4 y[t] == Exp[-t] Cos [2 t], t, s]
```

$$4\,\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] - s\,y[0] - y'[0] = \frac{1+s}{4+(1+s)^2}$$

```
step2 = step1 /. {y[0] → 0, y'[0] → -1}
```

$$1 + 4\,\text{LaplaceTransform}[y[t], t, s] + s^2\,\text{LaplaceTransform}[y[t], t, s] = \frac{1+s}{4+(1+s)^2}$$

```
step3 = Solve[step2, LaplaceTransform[y[t], t, s]]
```

$$\left\{\left\{\text{LaplaceTransform}[y[t], t, s] \to \frac{-4-s-s^2}{(4+s^2)\,(5+2\,s+s^2)}\right\}\right\}$$

```
solution = InverseLaplaceTransform[step3[[1, 1, 2]], s, t] // Expand
```

$$\left(\frac{1}{34}-\frac{2\,i}{17}\right) e^{(-1-2\,i)\,t} + \left(\frac{1}{34}+\frac{2\,i}{17}\right) e^{(-1+2\,i)\,t} - \frac{1}{17}\,\text{Cos}[2\,t] - \frac{4}{17}\,\text{Sin}[2\,t]$$

```
Plot[solution, {t, 0, 2 Pi}]
```



```
(* Der Plot zeigt wieder, dass die AW erfuellt sind *)

(* next -- komplizierteres Beispiel aus Mma-Hilfe *)

step1 = LaplaceTransform[y'''[t] - 5 y''[t] + 9 y'[t] - 5 y[t] ==
   DiracDelta''[t] + 2 DiracDelta '[t] + DiracDelta[t], t, s]
```

$-5\,\text{LaplaceTransform}[y[t], t, s] + s^3\,\text{LaplaceTransform}[y[t], t, s] + 9\,(s\,\text{LaplaceTransform}[y[t], t, s] - y[0]) - s^2\,y[0] - 5\left(s^2\,\text{LaplaceTransform}[y[t], t, s] - s\,y[0] - y'[0]\right) - s\,y'[0] - y''[0] == 1 + 2\,s + s^2$

```
step2 = step1 /. {y[0] → 0, y'[0] → 0, y''[0] -> 0} // FullSimplify
```

$(-1 + s)\,(5 + (-4 + s)\,s)\,\text{LaplaceTransform}[y[t], t, s] == (1 + s)^2$

```
step3 = Solve[step2, LaplaceTransform[y[t], t, s]] // ExpandDenominator
```

$\left\{\left\{\text{LaplaceTransform}[y[t], t, s] \to \frac{(1 + s)^2}{-5 + 9\,s - 5\,s^2 + s^3}\right\}\right\}$

```
solution = InverseLaplaceTransform[step3[[1, 1, 2]], s, t] // FullSimplify
```

$e^t\left(2 - e^t\,(\text{Cos}[t] - 7\,\text{Sin}[t])\right)$

```
Plot[solution, {t, 0.0, 0.2}, PlotRange -> All,
 PlotStyle → Thick, AxesLabel → {"t", y}, AxesOrigin → {0, 0}]
```




```
(*  Vergleich, man beachte, dass AW an der Stelle 0 nicht funktionieren: *)

Clear[y]
loes2 = DSolve[y'''[t] - 5 y''[t] + 9 y'[t] - 5 y[t] ==
     DiracDelta''[t] + 2 DiracDelta '[t] + DiracDelta[t] &&
    y[-1] == 0 && y'[-1] == 0 && y''[-1] == 0, y[t], t]  // FullSimplify
```

$\left\{\left\{y[t] \to -e^t\,\text{HeavisideTheta}[t]\left(-2 + e^t\,(\text{Cos}[t] - 7\,\text{Sin}[t])\right)\right\}\right\}$

```
(*   Die AW und die DeltaFunktionen stoeren sich so nicht gegenseitig.
  Weil:  AW wurden 'zurueckverlegt'.
        y≡0 wuerde ueberall eine Loesung sein,
wenn nicht die Deltas einen Ruck bei t=0 erzeugen wuerden.  *)
```

```
Plot[loes2[[1, 1, 2]], {t, -0.1, 0.23},
 PlotRange -> All, PlotStyle → Thick, AxesLabel → {"t", y}]
```




```
Plot[{solution, y[t] /. loes2}, {t, -0.1, 0.1}, PlotLegends → "Expressions"]
```




```
(*  Die Laplacetransformation geht ab t,s >0 los ,
wenn man die LaplaceLoesung fuer <0 einsetzt,
gibt es einen falschen Eindruck. Mit DSolve wird die korrekte Loesung mit
  einem Sprung angegeben. Fuer t>0 stimmen beide Loesungen ueberein. *)
```